

Calculations

Fredrik Meyer

September 14, 2016

Contents

1	Computations on dP6	1
1.1	Finding equations of deformations	1
1.2	Intersecting with two special hyperplanes	3
1.3	Intersecting with another set of hyperplanes	5
2	Toric mirrors	5
2.1	First example	5
2.2	Second example	7
3	Deformation calculations	7
3.1	Non-standard triangulation	7
4	Singularities	10
4.1	Of the toric hypersurface	10
5	Calabi-Yaus	11
5.1	Singular in a 5-dimensional toric	11

1 Computations on dP6

1.1 Finding equations of deformations

Consider the del Pezzo surface dP_6 of degree 6 embedded in \mathbb{P}^6 . Its ideal is defined as follows:

```
1 restart
S = QQ[x_1..x_6,y_0]
3 I = ideal(x_1*x_3-x_2*y_0,
           x_2*x_4-x_3*y_0,
```

```

5 | x_3*x_5-x_4*y_0,
  | x_4*x_6-x_5*y_0,
7 | x_5*x_1-x_6*y_0,
  | x_6*x_2-x_1*y_0,
9 | x_1*x_4-y_0^2,
  | x_2*x_5-y_0^2,
11| x_3*x_6-y_0^2)

```

We compute the two deformations of its affine cone using the package `VersalDeformations`.

```

1 | (F,R,G,C) = versalDeformation(gens I);
  | decompose ideal transpose mingens ideal G

```

The output are four lists of matrices entries in $\mathbb{Q}[\mathbf{x}] \otimes \mathbb{Q}[t_1, t_2, t_3]$. The list F consists of the equations of the family, and the list R of the relations. The list G gives equations for the base space. We have that F_0 is the matrix of generators of I , and that $F_i R_i \equiv 0 \pmod{t^{i+1}}$.

The decomposition of `ideal G` is the following:

```

i9 : decompose ideal transpose mingens ideal G
2 |
  | o9 = {ideal (t_1 - t_3), ideal (t_2 - t_3, t_1)}
4 |

```

Thus the base space splits into two components meeting transversely at the origin, of dimension 2 and 1, respectively. By doing a change of variables we can get rid of the linear terms:

```

1 | T = QQ[x_1..x_6,y_0,t_1,t_2,t_3,s_1,s_2,s_3];
  | gsub = sub(sub(ideal mingens ideal G,T), {t_2 => s_2+s_3-s_1,
  |       t_1 => s_3, t_3 => s_3-s_1})
3 | fsub = transpose sub(sub(sum F,T), {t_2 => s_2+s_3-s_1, t_1 =>
  |       s_3, t_3 => s_3-s_1})

```

Now the equations are easier:

```

1 | i13 : decompose gsub
3 | o13 = {ideal (s_1), ideal (s_3, s_2)}
  |

```

We can get equations for each of these families by setting $s_1 = 0$ and $s_3 = s_2 = 0$, respectively:

```

1 i25 : fsub1 = sub(fsub, s_1 => 0)
3 o25 = {-2} | x_1x_3-x_2y_0 |
        {-2} | x_2x_4-x_3y_0+x_3s_2+x_3s_3 |
5        {-2} | x_3x_5-x_4y_0+x_4s_3 |
        {-2} | x_4x_6-x_5y_0 |
7        {-2} | x_1x_5-x_6y_0+x_6s_2+x_6s_3 |
        {-2} | x_2x_6-x_1y_0+x_1s_3 |
9        {-2} | x_1x_4-y_0^2+y_0s_2+y_0s_3 |
        {-2} | x_2x_5-y_0^2+y_0s_2+2y_0s_3-s_2s_3-s_3^2 |
11       {-2} | x_3x_6-y_0^2+y_0s_3 |
13
13 o25 : Matrix T <--- T

```

And:

```

1 i26 : fsub2 = sub(fsub, {s_3 => 0, s_2 => 0})
2
3 o26 = {-2} | x_1x_3-x_2y_0 |
4        {-2} | x_2x_4-x_3y_0-x_3s_1 |
5        {-2} | x_3x_5-x_4y_0 |
6        {-2} | x_4x_6-x_5y_0-x_5s_1 |
7        {-2} | x_1x_5-x_6y_0 |
8        {-2} | x_2x_6-x_1y_0-x_1s_1 |
9        {-2} | x_1x_4-y_0^2-y_0s_1 |
10       {-2} | x_2x_5-y_0^2-y_0s_1 |
11       {-2} | x_3x_6-y_0^2-y_0s_1 |
12
14 o26 : Matrix T <--- T

```

1.2 Intersecting with two special hyperplanes

Consider dP_6 defined as above. Then consider the two hyperplanes

$$h_1 = x_1 + x_2 + x_3 + x_4 + x_5 + x_6$$

and

$$h_2 = x_1 - x_2 + x_3 - x_4 + x_5 - x_6.$$

We can compute the intersection with dP_6 in Macaulay2 as follows:

```

h1 = x_1+x_2+x_3+x_4+x_5+x_6
h2 = x_1-x_2+x_3-x_4+x_5-x_6
SS = S[r]/(r^2+3)
apply(decompose(sub(I + h1 + h2,SS)), j -> ideal mingens sub(j,
y_0 => 1))

```

The reason we create a new ring is that we need $\sqrt{-3}$ in order for the ideals to decompose. The results is the following list of ideals:

1. $(x_5 - 1, x_4 + x_6 + 1, x_3 + x_6 + 1, x_2 - 1, x_1 - x_6, r - 2x_6 - 1).$
2. $(x_5 - 1, x_4 + x_6 + 1, x_3 + x_6 + 1, x_2 - 1, x_1 - x_6, r + 2x_6 + 1).$
3. $(x_6 - 1, x_4 - x_5, x_3 - 1, x_2 + x_5 + 1, x_1 + x_5 + 1, r - 2x_5 - 1).$
4. $(x_6 - 1, x_4 - x_5, x_3 - 1, x_2 + x_5 + 1, x_1 + x_5 + 1, r + 2x_5 + 1).$
5. $(x_5 - x_6, x_4 - 1, x_3 + x_6 + 1, x_2 + x_6 + 1, x_1 - 1, r - 2x_6 - 1).$
6. $(x_5 - x_6, x_4 - 1, x_3 + x_6 + 1, x_2 + x_6 + 1, x_1 - 1, r + 2x_6 + 1).$

From this we can read off the coordinates in \mathbb{P}^6 :

Table 1: Table of singular points.

x_1	x_2	x_3	x_4	x_5	x_6	y_0
$\frac{-1+\sqrt{-3}}{2}$	1	$\frac{-1-\sqrt{-3}}{2}$	$\frac{-1-\sqrt{-3}}{2}$	1	$\frac{-1+\sqrt{-3}}{2}$	1
$\frac{-1+\sqrt{-3}}{2}$	1	$\frac{-1+\sqrt{-3}}{2}$	$\frac{-1+\sqrt{-3}}{2}$	1	$\frac{-1-\sqrt{-3}}{2}$	1
$\frac{-1-\sqrt{-3}}{2}$	$\frac{-1-\sqrt{-3}}{2}$	1	$\frac{-1+\sqrt{-3}}{2}$	$\frac{-1+\sqrt{-3}}{2}$	1	1
$\frac{-1+\sqrt{-3}}{2}$	$\frac{-1+\sqrt{-3}}{2}$	1	$\frac{-1-\sqrt{-3}}{2}$	$\frac{-1-\sqrt{-3}}{2}$	1	1
1	$\frac{-1-\sqrt{-3}}{2}$	$\frac{-1-\sqrt{-3}}{2}$	1	$\frac{-1+\sqrt{-3}}{2}$	$\frac{-1+\sqrt{-3}}{2}$	1
1	$\frac{-1+\sqrt{-3}}{2}$	$\frac{-1+\sqrt{-3}}{2}$	1	$\frac{-1-\sqrt{-3}}{2}$	$\frac{-1-\sqrt{-3}}{2}$	1

Note that the hexagonal group D_6 act transitively on the set of singular points.

1.3 Intersecting with another set of hyperplanes

Note that the group $\mathbb{Z}_2 \times \mathbb{Z}_2$ act on dP_6 by reflecting the triangulation in Figure 1a (in fact $\mathbb{Z}_2 \times \mathbb{Z}_2$ is a subgroup of D_6). $\mathbb{Z}_2 \times \mathbb{Z}_2$ is the full symmetry group of the triangulation in the figure.

2 Toric mirrors

2.1 First example

There are two examples of torics we have studied. The first one is this.

Let I and I' be the ideals of dP_6 with a disjoint set of variables. Let $J = I + I'$.

Remark. *This is a deformation of the Stanley-Reisner scheme corresponding to the join of two hexagons and two points.*

Then $Y_0 = \text{Proj}(S/J)$ is a toric variety whose associated polytope P has vertices the column of the following matrix:

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & -2 & 2 & 0 & -2 & 0 & 2 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -2 & -2 & 2 & 2 \\ -2 & 2 & 0 & -2 & 0 & 2 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -2 & -2 & 2 & 2 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & -1 & -1 & -1 & -1 & -1 & 1 & 1 & 1 & 1 & 1 & 1 \end{pmatrix} \quad (1)$$

The polytope P is reflexive with 87 lattice points. We can use the Batyrev-Borisov construction to find a toric mirror candidate to Y_0 .

The construction actually give two such families. Here's how: start with a reflexive polytope $P \subset M_{\mathbb{R}}$. A *nef partition* induces a decomposition V_1, V_2 of the vertex set of $P^\vee \subset N_{\mathbb{R}}$, such that $P^\vee = \text{conv}(V_1, V_2)$. Let $Q_i = \text{conv}(V_i)$. Then one forms $Q := Q_1 + Q_2 \subset N_{\mathbb{R}}$. Similarly, we have a dual nef partition, so that we can write $P = P_1 + P_2$.

A nef partition correspond to a complete intersection in X_P , and the dual nef partition correspond to a complete intersection in X_Q .

We can find (using SAGE for example) the polytopes P_1 and P_2 . They are given as follows:

$$P_1 = \text{conv} \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & -1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 1 & 1 & 0 & 0 & -1 \\ -1 & -1 & 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & 0 & 1 & 1 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 1 & 1 & 1 & 1 & 1 \end{pmatrix} \quad (2)$$

and

$$P_2 = \text{conv} \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & -1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 1 & 1 & 0 & 0 & -1 \\ -1 & -1 & 1 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & 0 & 1 & 1 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & -1 & -1 & -1 & -1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix} \quad (3)$$

There is a formula for the Q_i given the P_i :

$$Q_i = \{u \in N_{\mathbb{R}} \mid \langle u, m \rangle \geq -1 \text{ for all } m \in P_i, \langle u, m \rangle \geq 0 \forall m \in P_j, j \neq i\}$$

Then we find that Q_1 is given by

$$Q_1 = \text{conv} \begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ -1 & -1 & 0 & 0 & 0 & 1 & 1 & 1 \\ 0 & 1 & -1 & 0 & 1 & -1 & 0 & 0 \\ -1 & -1 & -1 & 0 & -1 & -1 & -1 & -1 \end{pmatrix} \quad (4)$$

and

$$Q_2 = \text{conv} \begin{pmatrix} -1 & -1 & 0 & 0 & 0 & 1 & 1 & 1 \\ 0 & 1 & -1 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 1 & 1 & 1 & 0 & 1 & 1 & 1 & 1 \end{pmatrix} \quad (5)$$

The polytope Q has 48 vertices:

[illegible]

The polar polytope Q^\vee is given by

$$\begin{pmatrix} -1 & -1 & -1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 1 & 1 \\ -1 & -1 & 0 & 0 & -1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 1 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & -1 & -1 & -1 & 0 & 0 & 0 & 0 & 1 & 1 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -1 & -1 & 0 & 0 & -1 & -1 & 1 & 1 & 0 & 0 & 1 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 1 & 0 & 1 & -1 & 0 & -1 & 0 & -1 & 0 & -1 & 0 & -1 & 0 & -1 & 0 & 0 & 1 & 0 & 1 & 0 & 1 \end{pmatrix}$$

One can use the program `nef_x` (written by Skarke et al) to compute the Hodge numbers corresponding to the nef partitions $P_1 + P_2$ and $Q_1 + Q_2$. It turns out that they are $(19, 19)$.

2.2 Second example

Again consider two copies of the homogeneous ideal of dP_6 . This time, however, embed them by mapping the origins in their polytopes to the same monomial. This is the same as letting $y_0 = y_1$ in the previous example, where y_i correspond to the origins of the polytopes. The corresponding polytope is just the product of two hexagons.

Remark. *This is a deformation of the Stanley-Reisner scheme corresponding to the join of two hexagons and a point.*


We get a 4-dimensional toric variety Y'_0 . Here, running `nef_x` gives Hodge numbers $(44, 8)$, giving an Euler characteristic of 72.

Remark. *This sounds promising, because earlier I've found a smoothing of this variety with Euler characteristic -72 .*


3 Deformation calculations

3.1 Non-standard triangulation

In Figure 1 are depicted two different triangulations of the hexagon. They correspond to different Gröbner bases of the ideal of dP_6 . Choosing the left triangulation turns out to give smaller-dimensional T^2 .



(a) The hexagon in another triangulation.



(b) The hexagon in the pulling triangulation.

Figure 1: Two triangulations of the hexagon.

Below are some deformation computations done on the Stanley-Reisner scheme corresponding to the join of two hexagons in this triangulation.

```

1 restart
  R1 = QQ[x_1..x_6,y_0]
3  R2 = QQ[z_1..z_6,y_1]

5  loadPackage "SimplicialComplexes"
  loadPackage "VersalDeformations"

7
  S1 = simplicialComplex {x_1*x_2*x_3, x_1*y_0*x_3, x_3*x_4*y_0,
    x_6*x_1*y_0, x_6*x_4*y_0, x_6*x_5*x_4}
9  S2 = simplicialComplex {z_1*z_2*z_3, z_1*y_1*z_3, z_3*z_4*y_1,
    z_6*z_1*y_1, z_6*z_4*y_1, z_6*z_5*z_4}
  I1 = ideal S1
11 I2 = ideal S2

13
  R = QQ[first entries vars R1 | first entries vars R2]
15 I = sub(I1,R) + sub(I2,R)

```

We first create the rings with the necessary variables, then we define the simplicial complexes by their maximal faces. Finally we create the ideal of the join, which is just the sum of the individual ideals.

We can compute T^2 :

```

1 i11 : CT^2(0, gens I);

3          115          76
o11 : Matrix R  <----- R

```

It is 76-dimensional, which is still quite big, but much smaller than if we had used the other triangulation.

We can compute T^1 :

```
i13 : T11 = CT^1(0, gens I);
2
18      58
4 o13 : Matrix R  <--- R
```

This is the module of first-order deformations, which is 58-dimensional.

Using the **VersalDeformations** package, we can lift the deformations to second order:

```
(f1,r1,g1,c1) = versalDeformation(gens I, T11, CT^2(0, gens I),
HighestOrder =>2, Verbose => 10);
```

Then `ideal sum g1` is the second order terms of the obstruction equations of the versal base space. The ideal is generated by 76 elements, and by inspection we see that half of them uses only half of the deformation parameters. This makes it easier to compute a decomposition of it. We do it as follows:

```
1 loadPackage "Binomials"
IG = ideal sum g1
3
IG1 = ideal ((IG_*)_ {0..37})
5 IG2 = ideal ((IG_*)_ {38..75})
7 T = QQ[t_1..t_27]
IGT1 = sub(IG1,T)
9 kList = BPD IGT1
11 T2 = QQ[t_28..t_58]
IGT2 = ideal mingens sub(IG,T2)
13 kList2 = BPD IGT2
```

This takes about a minute. We use the package **Binomials**, which decomposes binomial ideals fast. Finally, we put the decomposed parts in the same ring, and make a list of all the components:

```
1 TT = QQ[t_1..t_58]
kListTT = apply(kList, I -> sub(I,TT))
3 kList2TT = apply(kList2, I -> sub(I,TT))
5 intersect(intersect kListTT, intersect(kList2TT)) == IGTT ---
long time
```

```

7 comps = apply(toList ((set kListTT) ** (set kList2TT)), S ->
    ideal mingens (S#0 +S#1));
max apply(comps, dim) — answer: 44
9 maxComp = select(comps, I -> dim I == 44)

```

We check which component has the maximal dimension, and it turns out that it has dimension 44. We select that component in the last line. This ideal is generated by a subset of the deformation parameters.

We have been unable to lift this 44-dimensional family, though.

By selecting deformation parameters carefully, we can however find a family whose generic fiber is the ideal J above (the sum of two del Pezzo's). This family is 8-dimensional.

Remark. *This is very promising, because of the following: we have found a smoothing of the singular Calabi-Yau inside Y_0 , of which we computed the Euler characteristic to be -72 . This fits well with the Batyrev-Borisov calculations above: if the Hodge numbers are $(8, 44)$, we have potentially found an open subset of the moduli space of complex structures of this smoothing. The 8-dimensional subfamily could correspond to the space of complex structures on the mirror.*

4 Singularities

4.1 Of the toric hypersurface

Again consider the join of two hexagons, joined with a single vertex. This is a 4-dimensional simplicial complex, which deforms to the toric variety with polytope with vertices $(v, 0)$ and $(0, v)$, where v are vertices of the hexagon. This is the same toric variety as in “Second example” above.

If JJ is the ideal of this toric variety, the following code will compute its singularities.

```

1 singlist = {}
  for i from 1 to 6 do {
3     sz = sub(JJ, z_i == 1);
     sx = sub(JJ, x_i == 1);
5     singz = radical ideal mingens ideal singularLocus
        minimalPresentation sz;
     singx = radical ideal mingens ideal singularLocus
        minimalPresentation sx;
7     singsx = (decompose singx);

```

```

9      singsz = (decompose singz);
      sings = singsx | singsz;
      invz = sz.cache.minimalPresentationMap;
11     invx = sx.cache.minimalPresentationMap;
      singlist = singlist | apply(singsx, I -> homogenize(preimage
      (invx, singx), x_i)) | apply(singsz, I -> homogenize(preimage(
      invz, singz), z_i));
13     }
      sings = intersect singlist

```

X_P have 48 curves as singularities. Hence a general hyperplane section will have 48 isolated singularities.

Remark. *Local calculations hint that these should be nodal singularities, but I don't immediately know how to see this.*

5 Calabi-Yaus

5.1 Singular in a 5-dimensional toric

Let Y_0 be the toric variety from embedded in \mathbb{P}^{13} by (a subset of) the anticanonical system (use only the vertices and the origin). It is the join of two del Pezzo surfaces. The singular locus is the disjoint union of these del Pezzo surfaces. If we intersect Y_0 with two generic sections of the anticanonical bundle (in other words: two generic sections of $\mathcal{O}_{Y_0}(1)$), we get a Calabi-Yau with isolated singularities.

In fact, local calculations on separate del Pezzos show that we don't need the hyperplanes to be generic. We'd like to have a Calabi-Yau on which a group act, so we want invariant hyperplanes. From Section , we know that $\mathbb{Z}_2 \times \mathbb{Z}_2$ act on each del Pezzo on Y_0 . Hence we have a action of \mathbb{Z}_2^5 on Y_0 , given by permuting each factor, and exchanging the factors.

Remark. *We also have an action of $D_6 \times D_6 \times \mathbb{Z}_2$ on Y_0 , but since we want the action to descend to the Calabi-Yau, we'd like invariant hyperplanes. However, this group is too big - there is only one invariant hyperplane with respect to this group (the hyperplane with all coefficients equal).*

There is a 3-dimensional vector space of invariant hyperplanes from $\mathcal{O}_{Y_0}(1)$. We choose two of them:

$$x_1 + x_2 + x_3 + x_4 + x_5 + x_6 + y_0 + z_1 + z_2 + z_3 + z_4 + z_5 + z_6 + y_1 = 0$$

and

$$x_1 + x_2 + 2x_3 + x_4 + x_5 + 2x_6 + y_0 + z_1 + z_2 + 2z_3 + z_4 + z_5 + 2z_6 + y_1 = 0.$$

Local calculations show that this choice of hyperplanes give isolated singularities. They are all of the form $C(dP_6)$ and there are 12 of them. They come in two orbits under the group action: 4 with a \mathbb{Z}_2 stabilizer, and 8 with trivial stabilizer. The four with nontrivial stabilizer have coordinates

$$\begin{aligned} & (x_1 : x_2 : x_3 : x_4 : x_5 : x_6 : y_0 : z_1 : z_2 : z_3 : z_4 : z_5 : z_6 : z_1) \\ &= (0 : 0 : 0 : 1 : -1 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0), \\ &= (1 : -1 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0), \\ &= (0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 0 : 1 : -1 : 0 : 0), \\ &= (0 : 0 : 0 : 0 : 0 : 0 : 0 : 1 : -1 : 0 : 0 : 0 : 0 : 0). \end{aligned}$$

The other 8 have all nonzero coefficients with irrational coordinates. The exact coordinates can be computed symbolically in Mathematica for example.